Tema 7
Cálculo diferencial de funciones reales de variable real (1)

Objetivos:

1. Calcular la derivada de una función con wxMaxima.
2. Calcular derivadas de orden superior.
3. Calcular la diferencial de una función.
4. Calcular derivadas de funciones definidas implícitamente.

Contenidos:

07-1. Derivada de una función. Funciones derivables. Cálculo de derivadas.
07-2. Reglas de derivación. Derivadas de orden superior.
07-3. Diferencial de una función. Aplicaciones.
07-4. Funciones implícitas. Derivación de funciones implícitas.

Referencias

AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.

APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.

AP86 APOSTOL, T.M. (1986)
Análisis Matemático

BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima

BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).

ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.

ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable

GV07a GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 1: números y ecuaciones

GV07b GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 2: límites y derivadas

JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.

RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima

RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería

RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.

SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).

VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

07-1.- Derivada de una función. Funciones derivables. Cálculo de derivadas

Los contenidos de este apartado se desarrollan en el archivo **Tema_07-1.wxm**.

Si $f: A \subset \mathbb{R} \to \mathbb{R}$ es una función real de variable real y $a \in A$ es un punto interior del campo de existencia de la función, se llama derivada de la función en este punto, el siguiente límite, si existe

$$Df(a) = \lim_{h \to 0} \frac{f(a+h) - f(a)}{h} = \lim_{x \to a} \frac{f(x) - f(a)}{x - a}$$

La interpretación geométrica de este concepto es bien conocida: es la pendiente de la recta tangente a la gráfica de la función en el punto $(a, f(a))$ cuando esta recta existe, expresada como el límite de las pendientes de las rectas secantes a esta gráfica que pasan por los puntos $(a, f(a))$.

Por ejemplo, si se considera la función definida por:

$$f(x) = x^2 - 2x + 2, \ \ x \in \mathbb{R}$$

entonces a continuación se puede ver esta interpretación geométrica con unas rectas secantes que pasan por el punto $(2, f(2))$.

Implementando los cálculos con wxMaxima para la representación gráfica, se obtiene:

```
(%i1) f1(x):=x^2-2*x+2$
      y1(x):=4*x-6$    y2(x):=3*x-4$    y3(x):=2*x-2$
```

```
--> wxplot2d([f1(x),y1(x),y2(x),y3(x)], [x,1,4], [legend, "f1(x)", "y1(x)", "y2(x)", "y3(x)"] ,
[style, [lines, 2]] , [gnuplot_preamble, "set grid"], [nticks, 200] )$
```

(%t5)



Con wxMaxima se puede calcular la derivada de una función en un punto aplicando la definición anterior:

```
(%i6) 'limit((f1(x)-f1(2))/(x-2), x, 2)=limit((f1(x)-f1(2))/(x-2), x, 2);
```

(%o6) $\lim_{x\to 2} \frac{x^2-2x}{x-2}=2$

Si una función tiene derivada en un punto es dice derivable o diferenciable en este punto y se dice no derivable o no diferenciable, en caso contrario. A continuación se muestra un ejemplo de función no derivable en un punto.

```
(%i7) f2(x):=abs(x^2-2*x-8);
```

(%o7) $f2(x):=|x^2-2x-8|$

```
(%i8) wxplot2d([f2(x)], [x,0, 8],[y,-2,12] , [legend, "f2(x)"] , [ylabel, ""], [style, [lines, 2]] ,
      [gnuplot_preamble, "set grid"] )$
plot2d: some values were clipped.
```

(%t8)



En el caso de la función $f(x)=|x^2-2x-8|$ representada en la gráfica anterior y si se considera el punto a = 4, la respuesta de wxMaxima es:

```
(%i9) 'limit((f2(x)-f2(4))/(x-4), x, 4)=limit((f2(x)-f2(4))/(x-4), x, 4);
```

(%o9) $\lim_{x\to 4} \frac{|x^2-2x-8|}{x-4}=und$

La conclusión es que la función no es derivable en este punto. Entonces se pueden calcular las derivadas laterales, designadas $D_+f(a)$ y $D_-f(a)$, que son los límites laterales de la expresión de la definición de derivada de una función en un punto. Si la función es derivable ambos límites existen y son iguales y si la función no es derivable, entonces o bien las derivadas laterales existen y no son iguales o bien alguno de los límites no existe. En el caso de la función considera antes, las derivadas laterales son:

```
(%i10) '(D-f2(4)='limit((f2(x)-f2(4))/(x-4), x, 4, minus)) =
                 limit((f2(x)-f2(4))/(x-4), x, 4, minus);
```

(%o10) $D\text{-}f2(4)= \lim\limits_{x \to 4-} \dfrac{f2(x)-f2(4)}{x-4} = -6$

```
(%i11) '(D+f2(4)='limit((f2(x)-f2(4))/(x-4), x, 4, plus)) =
                 limit((f2(x)-f2(4))/(x-4), x, 4, plus);
```

(%o11) $D+f2(4)= \lim\limits_{x \to 4+} \dfrac{f2(x)-f2(4)}{x-4} = 6$

Cálculo de la derivada de una función

Para calcular la derivada de una función en un punto cualquiera con wxMaxima, es decir, para calcular la llamada función derivada, hay que aplicar la instrucción "diff" indicando entre paréntesis el símbolo asignado a la función que debe haber sido definida previamente y la variable respecto a la cual se deriva. Por ejemplo, si se ha definido $f_1(x) = x^2 - 2x + 2$ entonces:

```
(%i12) Df1(x)=diff(f1(x), x);
```

(%o12) $Df1(x) = 2\,x-2$

Si se quiere calcular la derivada en un punto concreto, es decir, el valor de la función derivada en este punto, se podría pensar que la metodología es similar a la que permite el cálculo del límite de una función en un punto. Así, si se quiere calcular la derivada de la función anterior en el punto a = 3, se podría pensar que la instrucción a aplicar es:

```
(%i13) Df1(3)=diff(f1(x), x, 3);
```

(%o13) $Df1(3) = 0$

resultado que no es correcto ya que $Df_1(3) = 4$. Así, pues, esta instrucción no es la correcta para este fin y es necesario implementar una estrategia adecuada. Una opción plausible es aplicar la instrucción "define" para definir la función derivada y aplicar esta función en el punto concreto. Así, en el ejemplo anterior, se puede proceder de la manera siguiente:

```
(%i14) define ( Df1(x) , diff(f1(x), x) );
```

(%o14) $Df1(x) := 2\,x-2$

Ahora se puede calcular la derivada en un punto concreto o en un punto cualquiera:

(%i15) 'Df1(3) = Df1(3);   'Df1(a) = Df1(a);

(%o15) $Df1(3)=4$

(%o16) $Df1(a)=2\,a-2$

Hay que tener cuidado con esta metodología. Por ejemplo, si se considera la función definida por:

$$F(x)=\begin{cases} x^2 \sin\left(\dfrac{1}{x}\right), & \text{si } x\neq 0 \\ 0, & \text{si } x=0 \end{cases}$$

y ahora se calcula la derivada con wxMaxima, resulta

(%i17) F(x):=x^2*(sin(1/x));   diff(F(x),x);

(%o17) $F(x):=x^2\sin\left(\frac{1}{x}\right)$

(%o18) $2\sin\left(\frac{1}{x}\right)x-\cos\left(\frac{1}{x}\right)$

Por tanto, si se define la función derivada mediante:

(%i19) define (DF(x) , diff(F(x),x));

(%o19) $DF(x):=2\sin\left(\frac{1}{x}\right)x-\cos\left(\frac{1}{x}\right)$

y ahora se quiere calcular la derivada en el origen como el valor en este punto de la función, está claro que esto no se puede hacer ya que la expresión anterior no está definida en el origen. En este caso se puede ver que la función considerada es derivable en el origen y se cumple $DF(0)=0$; en efecto:

(%i20) 'DF(0)=limit((F(x))/(x), x, 0);

(%o20) $DF(0)=0$

Derivada infinita.

Otro caso a considerar es la posibilidad de que el límite de definición de la derivada no sea finito y los límites laterales existan y coincidan. Tal es el caso, por ejemplo de la función definida por:

$$g(x)=\begin{cases} -\sqrt{-x}, & \text{si } x\leq 0 \\ \sqrt{x}, & \text{si } x>0 \end{cases}$$

(%i21) g1(x):=-sqrt(-x);   g2(x):=sqrt(x);
g(x):= if x<=0 then g1(x) else g2(x);

(%o21) $g1(x):=-\sqrt{-x}$

(%o22) $g2(x):=\sqrt{x}$

(%o23) $g(x):=\text{if } x<=0 \text{ then } g1(x) \text{ else } g2(x)$

(%i24) wxplot2d([g(x)], [x,-2,2], [y, -2,2], [legend, "g1(x)"] , [ylabel, ""], [style, [lines, 2]] , [gnuplot_preamble, "set grid"], [nticks, 200])$

(%t24)



Si se calcula la derivada de cada una de las expresiones analíticas de la función, resulta

(%i25) diff(g1(x), x);   diff(g2(x), x);

(%o25) $\dfrac{1}{2\sqrt{-x}}$

(%o26) $\dfrac{1}{2\sqrt{x}}$

Las derivadas de la función no están definidas en el origen, por tanto se han de calcular las derivadas laterales en este punto aplicando la definición de derivada lateral:

(%i27) '(D-g(0)) = limit((g1(x)-g1(0))/(x), x, 0, minus);
'(D+g(0)) = limit((g2(x)-g2(0))/(x), x, 0, plus);

(%o27) $D-g(0)=\infty$

(%o28) $D+g(0)=\infty$

En conclusión se puede afirmar que $Dg(0)=+\infty$ (extensión de la derivada a la recta real ampliada).

Para terminar este apartado mostramos que con wxMaxima se puede calcular la función derivada de una función que tenga parámetros en su expresión analítica. Así, por ejemplo:

```
(%i29) f4(x):=a_0 + a_1*x + a_2*x^2 + a_3*x^3;
       Df4(x) = diff(f4(x), x);
```

(%o29) $f4(x):=a\_0+a\_1\,x+a\_2\,x^{2}+a\_3\,x^{3}$

(%o30) $Df4(x)=3\,a\_3\,x^{2}+2\,a\_2\,x+a\_1$

```
(%i31) f5(x):=exp(a*x);
       Df5(x) = diff(f5(x), x);
```

(%o31) $f5(x):=\exp(a\,x)$

(%o32) $Df5(x)=a\,\%e^{a\,x}$

```
(%i33) f6(x):=sin(a*x)+x*log(b*x);
       Df6(x) = diff(f6(x), x);
```

(%o33) $f6(x):=\sin(a\,x)+x\,\log(b\,x)$

(%o34) $Df6(x)=\log(b\,x)+a\,\cos(a\,x)+$

07-2. Reglas de derivación. Derivadas de orden superior.

Los contenidos de este apartado se desarrollan en el archivo **Tema_07-2.wxm**.

Les reglas de derivación, es decir, la relación entre la derivación y las operaciones elementales con funciones, se pueden escribir con wxMaxima, tal como se muestra a continuación. Obsérvese que se utiliza la notación clásica pera la derivada de una función y no la más moderna con el operador derivación *D*.

Derivada de la suma y diferencia de funciones:

```
(%i1) 'diff(f(x)+g(x), x)=diff(f(x)+g(x), x);
      'diff(f(x)-g(x), x)=diff(f(x)-g(x), x);
```

$$(\%o1)\ \frac{d}{d\,x}(g(x)+f(x))=\frac{d}{d\,x}g(x)+\frac{d}{d\,x}f(x)$$

$$(\%o2)\ \frac{d}{d\,x}(f(x)-g(x))=\frac{d}{d\,x}f(x)-\frac{d}{d\,x}g(x)$$

Derivada del producto para una constante y del producto de dos funciones:

```
(%i3) 'diff(a*f(x), x)=diff(a*f(x), x);
      'diff(f(x)*g(x), x)=diff(f(x)*g(x), x);
```

$$(\%o3)\ \frac{d}{d\,x}(a\,f(x))=a\left(\frac{d}{d\,x}f(x)\right)$$

$$(\%o4)\ \frac{d}{d\,x}(f(x)\,g(x))=f(x)\left(\frac{d}{d\,x}g(x)\right)+g(x)\left(\frac{d}{d\,x}f(x$$

Derivada del cociente de dos funciones:

```
(%i5) 'diff(f(x)/g(x), x)=diff(f(x)/g(x), x);
```

$$(\%o5)\ \frac{d}{d\,x}\frac{f(x)}{g(x)}=\frac{\frac{d}{d\,x}f(x)}{g(x)}-\frac{f(x)\left(\frac{d}{d\,x}g(x)\right)}{g(x)^2}$$

Se observa que la expresión de derivada del cociente de ambas funciones no está escrita en la forma habitual, para conseguirla, simplificamos la expresión anterior:

```
(%i6) ratsimp(%);
```

$$(\%o6)\ \frac{d}{d\,x}\frac{f(x)}{g(x)}=-\frac{f(x)\left(\frac{d}{d\,x}g(x)\right)-g(x)\left(\frac{d}{d\,x}f(x)\right)}{g(x)^2}$$

Derivada de la potencia de una función con exponente una función:

```
(%i7) 'diff(f(x)^(g(x)), x)=diff(f(x)^(g(x)), x);
```

$$(\%o7)\ \frac{d}{d\,x}\,f(x)^{g(x)}=f(x)^{g(x)}\left(\log(f(x))\left(\frac{d}{d\,x}\,g(x)\right)+\frac{g(x)\left(\frac{d}{d\,x}\,f(x)\right)}{f(x)}\right)$$

Esta expresión se puede operar para obtener la expresión habitual de la derivada de una potencia de funciones:

```
(%i8) expand(%);
```

$$(\%o8)\ \frac{d}{d\,x}\,f(x)^{g(x)}=f(x)^{g(x)}\log(f(x))\left(\frac{d}{d\,x}\,g(x)\right)+f(x)^{g(x)-1}\,g(x)\left(\frac{d}{d\,x}\,f(x)\right)$$

Derivada de la composición de dos funciones:

```
(%i9) 'diff(f(g(x)), x)=diff(f(g(x)), x);
```

$$(\%o9)\ \frac{d}{d\,x}\,f(g(x))=\frac{d}{d\,x}\,f(g(x))$$

Se puede observar que la derivada de la función compuesta (regla de la cadena) no se desarrolla como se podía esperar. Sin embargo, no hace falta deducir que con wxMaxima no se pueden calcular derivadas de funciones compuestas, se puede hacer y en el ejemplo que sigue se mostrará la metodología para hacerlo.

Veamos un ejemplo de la aplicación de las expresiones anteriores. Consideremos, por ejemplo, las funciones definidas por:

$$f(x)=1+x^2,\ x\in\mathbb{R};\quad g(x)=\sin(2x+1),\ x\in\mathbb{R}$$

```
(%i10) f(x):=1+x^2;
       g(x):=sin(2*x-1);
```

$$(\%o10)\ f(x):=1+x^2$$

$$(\%o11)\ g(x):=\sin(2\,x-1)$$

La derivada de la suma y de la diferencia de estas funciones es:

```
(%i12) 'diff(f(x)+g(x), x)=diff(f(x)+g(x), x);
```

$$(\%o12)\ \frac{d}{d\,x}(\sin(2\,x-1)+x^2+1)=2\cos(2\,x-1)+2\,x$$

```
(%i13) 'diff(f(x)-g(x), x)=diff(f(x)-g(x), x);
```

$$(\%o13)\ \frac{d}{d\,x}(-\sin(2\,x-1)+x^2+1)=2\,x-2\cos(2\,x-1)$$

La derivada del producto y del cociente de estas funciones es:

```
(%i14) 'diff(f(x)*g(x), x)=diff(f(x)*g(x), x);
```

$$(\%o14)\ \frac{d}{d\,x}\left((x^2+1)\sin(2\,x-1)\right)=2\,x\sin(2\,x-1)+2\,(x^2+1)\cos(2\,x-1)$$

```
(%i15) 'diff(f(x)/g(x), x)=diff(f(x)/g(x), x);
```

$$(\%o15)\ \frac{d}{d\,x}\frac{x^2+1}{\sin(2\,x-1)}=\frac{2\,x}{\sin(2\,x-1)}-\frac{2\,(x^2+1)\cos(2\,x-1)}{\sin(2\,x-1)^2}$$

Tal como se ha comentado antes, simplificamos la expresión de la derivada del cociente:

```
(%i16) ratsimp(%);
```

$$(\%o16)\ \frac{d}{d\,x}\frac{x^2+1}{\sin(2\,x-1)}=\frac{2\,x\sin(2\,x-1)+(-2\,x^2-2)\cos(2\,x-1)}{\sin(2\,x-1)^2}$$

La derivada de la potencia de estas funciones es:

```
(%i17) 'diff(f(x)^(g(x)), x)=diff(f(x)^(g(x)), x);
```

$$(\%o17)\ \frac{d}{d\,x}(x^2+1)^{\sin(2\,x-1)}=(x^2+1)^{\sin(2\,x-1)}\left(2\cos(2\,x-1)\log(x^2+1)+\frac{2\,x\sin(2\,x-1)}{x^2+1}\right)$$

Y con la simplificación comentada:

```
(%i18) expand(%);
```

$$(\%o18)\ \frac{d}{d\,x}(x^2+1)^{\sin(2\,x-1)}=2\,(x^2+1)^{\sin(2\,x-1)}\cos(2\,x-1)\log(x^2+1)+2\,x\,(x^2+1)^{\sin(2\,x-1)-1}\sin(2\,x-1)$$

La derivada de les funciones compuestas $(f\circ g)$ i $(g\circ f)$ es:

```
(%i19) 'diff(f(g(x)), x)=diff(f(g(x)), x);
```

$$(\%o19)\ \frac{d}{d\,x}(\sin(2\,x-1)^2+1)=4\cos(2\,x-1)\sin(2\,x-1)$$

```
(%i20) 'diff(g(f(x)), x)=diff(g(f(x)), x);
```

$$(\%o20)\ \frac{d}{d\,x}\sin(2\,(x^2+1)-1)=4\,x\cos(2\,(x^2+1)-1)$$

Derivadas de orden superior.

Les derivadas de orden superior, o derivadas sucesivas, se definen mediante:

$$D^k f(x)=D\left(D^{k-1}f(x)\right),\ \ k\in\mathbb{N}$$

Estas derivadas de orden superior se calculan de forma sencilla con wxMaxima: la sintaxis sólo requiere poner después de la variable un número natural que indica el número de veces que se deriva la función.

Per ejemplo, si se considera la función polinómica $P(x)=4x^3-5x^2+2x-6$, la derivada (de primer orden) se puede calcular mediante:

```
(%i1) P1(x):=4*x^3-5*x^2+2*x-6;
```

(%o1) $P1(x):=4x^3-5x^2+2x-6$

```
(%i2) 'diff(P1(x), x, 1)=diff(P1(x), x, 1);
```

(%o2) $\frac{d}{dx}(4x^3-5x^2+2x-6)=12x^2-10x+2$

```
(%i3) 'diff(P1(x), x)=diff(P1(x), x);
```

(%o3) $\frac{d}{dx}(4x^3-5x^2+2x-6)=12x^2-10x+2$

Se observa pues que si después de la variable no se pone ninguna indicación, por defecto se calcula la derivada primera (o de orden 1). Si se quiere escribir la derivada con la notación del operador derivación, entonces se puede hacer lo siguiente:

```
(%i4) DP1(x)=diff(P1(x), x);
```

(%o4) $DP1(x)=12x^2-10x+2$

Para las derivadas de orden superior 2, 3 y 4 (cualquier otra derivada de orden superior a 4 es nula), la salida del programa es:

```
(%i5) 'diff(P1(x), x, 2)=diff(P1(x), x, 2);
```

(%o5) $\frac{d^2}{dx^2}(4x^3-5x^2+2x-6)=24x-10$

```
(%i6) 'diff(P1(x), x, 3)=diff(P1(x), x, 3);
```

(%o6) $\frac{d^3}{dx^3}(4x^3-5x^2+2x-6)=24$

```
(%i7) 'diff(P1(x), x, 4)=diff(P1(x), x, 4);
```

(%o7) $\frac{d^4}{dx^4}(4x^3-5x^2+2x-6)=0$

Utilizando la notación del operador derivación:

```
(%i8) D2P1(x)=diff(P1(x), x, 2);
```

(%o8) $D2P1(x)=24x-10$

```
(%i9) D3P1(x)=diff(P1(x), x, 3);
(%o9) D3P1(x)= 24
(%i10) D4P1(x)=diff(P1(x), x, 4);
(%o10) D4P1(x)= 0
```

Para poder calcular derivadas de orden superior aplicando la derivación a la función derivada de orden anterior, hay que aplicar la instrucción "define" para poder tener definida la derivada funcional y que el programa pueda trabajar con esta función. Así, por ejemplo, en el caso del polinomio anterior:

```
(%i11) define (DP1(x) , diff(P1(x), x));
(%o11) DP1(x):= 12 x^2 - 10 x + 2
(%i12) define (D2P1(x) , diff(P1(x), x, 2));
(%o12) D2P1(x):= 24 x - 10
(%i13) define (D3P1(x) , diff(P1(x), x, 3));
(%o13) D3P1(x):= 24
(%i14) define (D4P1(x) , diff(P1(x), x, 4));
(%o14) D4P1(x):= 0
```

07-3. Diferencial de una función. Aplicaciones

Los contenidos de este apartado se desarrollan en el archivo **Tema_07-3.wxm**.

Si $f : A \subset \mathbb{R} \to \mathbb{R}$ es una función real de variable real diferenciable en un punto $a \in A$ interior del campo de existencia de la función, se llama diferencial de *f* en el punto *a*, designada df_a, la aplicación lineal de $\mathbb{R}$ en $\mathbb{R}$ definida por

$$df_a(h) = Df(a)h, \ \ h \in \mathbb{R}$$

Por ejemplo, se considera la función diferenciable

$$f(x) = x^2 - 2x + 2, \ \ x \in \mathbb{R}$$

Definámosla con wxMaxima y calculemos su función derivada:

```
(%i1) f(x):=x^2-2*x+2;
(%o1) f(x):=x^2-2 x+2
(%i2) define (Df(x) , diff(f(x), x) );
(%o2) Df(x):=2 x-2
```

Entonces, la diferencial de esta función en un punto $a \in \mathbb{R}$ es:

```
(%i3) define (dfa(h), Df(a) * h);
(%o3) dfa(h):=(2 a-2) h
```

La diferencial es la única aplicación lineal que aproxima localmente los incrementos de la función en un entorno del punto, es decir:

$$df_a(h) \approx \Delta f(a;h) = f(a+h) - f(a)$$

En el caso anterior el incremento en un punto es:

```
(%i4) Increm_f_a(h)=f(a+h)-f(a);
(%o4) Increm_f_a(h)=(h+a)^2-2 (h+a)-a^2+2 a
(%i5) expand(%);
(%o5) Increm_f_a(h)=h^2+2 a h-2 h
(%i6) Increm_f_a(h):=h^2+2*a*h-2*h;
(%o6) Increm_f_a(h):=h^2+2 a h+(-2) h
```

La propiedad mencionada de la diferencial se expresa entonces:

```
(%i6) Increm_f_a(h):=h^2+2*a*h-2*h;
```

$$\text{(\%o6) Increm\_f\_a}(h):=h^2+2\,a\,h+(-2)\,h$$

```
(%i7) '(Increm_f_a(h)-dfa(h))=Increm_f_a(h)-dfa(h);
```

$$\text{(\%o7) Increm\_f\_a}(h)-\text{dfa}(h)=h^2-(2\,a-2)\,h+2\,a\,h-2\,h$$

```
(%i8)  expand(%);
```

$$\text{(\%o8) Increm\_f\_a}(h)-\text{dfa}(h)=h^2$$

La propiedad esencial de la diferencial se ha aplicado clásicamente a cálculos aproximados que la existencia actualmente de herramientas potentes de cálculo ha dejado casi obsoletos. Sin embargo, haremos a continuación un ejemplo de aplicación de la diferencial de una función que en ocasiones puede tener interés.

Se trata de establecer el subconjunto de puntos de la recta real en el que se puede aproximar la expresión $(x+2.15)^{1/3}$ por $x^{1/3}$ con un error inferior a 10^{-4}. Si se considera la función $f(x)=x^{1/3}$, diferenciable en toda la recta real exceptuando el origen, entonces se cumple

```
(%i1) f(x):=x^(1/3);
```

$$\text{(\%o1) } f(x):=x^{1/3}$$

```
(%i2) define (Df(x) , diff(f(x), x) );
```

$$\text{(\%o2) Df}(x):=\frac{1}{3\,x^{2/3}}$$

Por lo tanto la función diferencial esta definida mediante

```
(%i3) define (dfa(h), Df(a) * h);
```

$$\text{(\%o3) dfa}(h):=\frac{h}{3\,a^{2/3}}$$

El valor de la diferencial en h = 2.15 es una aproximación del incremento debido a este desplazamiento, es decir:

```
(%i4) dfa(2.15);
```

$$\text{(\%o4) }\frac{0.71666666666667}{a^{2/3}}$$

Per lo tanto, se trata de encontrar los puntos que hacen que este incremento no supere la cota fijada, hecho que se expresa mediante la desigualdad:

(%i5) 0.71666666666667/a^(2/3)<10^(-4);

$$(\%o5)\ \frac{0.71666666666667}{a^{2/3}}<\frac{1}{10000}$$

Resolviendo esta inecuación se obtiene:

(%i6) a>(0.71666666666667*10000)^(3/2);

(%o6) a > 606702.5325173516

Así pues, la condición se cumple en el intervalo]606702.53252, +∞[.

07-4. Funciones implícitas. Derivación de funciones implícitas

Los contenidos de este apartado se desarrollan en el archivo **Tema_07-4.wxm**.

Las funciones implícitas de una variable o funciones definidas implícitamente son aquellas que se definen mediante ecuaciones de curvas planas en las que intervienen dos variables. Una curva plana es un conjunto de puntos del plano definidos por una ecuación del tipo $F(x,y)=0$. M Mediante esta ecuación se puede definir implícitamente una función $x \mapsto y(x)$ tal que su imagen en un punto x es el valor $y(x)$ tal que $F(x,y(x))=0$. Por ejemplo, la curva plana de ecuación $x^3+y^3=1$ define implícitamente una función $x \mapsto y(x)$ tal que la imagen es el valor tal que $x^3+y(x)^3=1$. Los detalles sobre este tema y las propiedades de las funciones definidas implícitamente, se pueden encontrar en las referencias.

Para representar gráficamente con wxMaxima una curva plana hay que cargar primero el paquete llamado "draw2d" y utilizar la instrucción "implicit", especificando la ecuación de la curva y los rangos de las variables, tal como se verá a continuación. El gráfico resultante se puede modificar con las opciones correspondientes, utilizando el cuadro de diálogo al efecto. Si se quiere obtener la gráfica "pegada" en el mismo fichero de cálculos, entonces hay que utilizar la instrucción "wxdraw2d".

Así, la representación gráfica de la curva plana $x^3+y^3=1$ se obtendría de la manera siguiente:

```
(%i1) load(draw)$
      wxdraw2d(
      grid = true,   line_type = solid,  line_width = 2, color = blue,
      implicit(x^3+y^3-1=0, x, -3,3, y, -3,3));
```

(%t2)



(%o2)

Veamos otro ejemplo en el que se cambia el color empleado en la gráfica de la curva plana; se trata de la curva plana de ecuación $x^3 - xy + y^3 = 1$:

```
(%i3) wxdraw2d(
      grid = true,    line_type = solid,  line_width = 2, color = red,
      implicit(x^3-x*y+y^3-1=0, x, -2,2, y, -2,2));
```

(%t3)



(%o3)

Derivación implícita.

Si una curva plana define implícitamente una función diferenciable, se pueden calcular las derivadas de la función definida implícitamente. Veamos a continuación con un ejemplo la metodología y la sintaxis a aplicar. Consideramos la ecuación, la cual define una curva plana representada en la figura 07-3 (b).

En primer lugar hay que escribir la ecuación de referencia mediante la cual se relacionan inicialmente las variables:

```
(%i1) Eq1:x^3-x*y+y^3-1=0;
```

(%o1) $y^3 - x\,y + x^3 - 1 = 0$

A continuación hay que especificar cuál de las variables es la que se considera función definida implícitamente y cuál es la que sigue siendo la variable. La instrucción de wxMaxima es “depends” con un paréntesis en el que se ponen las variables ordenadas: en primer lugar la “función” y en segundo lugar la “variable”, en este caso la sintaxis es

```
(%i2) depends(y,x);
```

(%o2) [y(x)]

A partir de la aplicación de esta instrucción, el programa interpreta que y es una función de x definida mediante una ecuación. Si la función definida implícitamente es diferenciable, se puede calcular la derivada de esta función derivando en la ecuación inicial, mediante la instrucción:

```
(%i3) diff(Eq1,x);
```

$$(\%o3)\ 3\,y^2\left(\frac{d}{d\,x}\,y\right)-x\left(\frac{d}{d\,x}\,y\right)-y+3\,x^2=0$$

En esta ecuación se puede ver como wxMaxima interpreta que se ha asignado el rol funcional *y*(*x*) y se ha derivado la ecuación inicial considerando *x* la variable y *y* una función de *x*. Se puede designar la ecuación anterior con una referencia, por ejemplo

```
(%i4) 'Eq2=Eq2:%;
```

$$(\%o4)\ Eq2=\left(3\,y^2\left(\frac{d}{d\,x}\,y\right)-x\left(\frac{d}{d\,x}\,y\right)-y+3\,x^2=0\right)$$

Esta ecuación se puede reescribir utilizando la notación de la derivada con el operador derivación:

```
(%i5) Eq2b:3*y^2*Dy(x)-x*Dy(x)-y+3*x^2=0;
```

$$(\%o5)\ 3\,Dy(x)\,y^2-y-x\,Dy(x)+3\,x^2=0$$

Si se quiere calcular la ecuación de la recta tangente a la curva definida por la ecuación Eq1 en un punto, por ejemplo el punto (1,0), hay que sustituir estos valores en la ecuación Eq2, obteniéndose:

```
(%i6) Eq3:-1*('diff(y,x,1))-0+3*1^2=0;
```

$$(\%o6)\ 3-\frac{d}{d\,x}\,y=0$$

Por lo tanto se cumple

```
(%i7) 'Dy(1)=3;
```

$$(\%o7)\ Dy(1)=3$$

La ecuación de la recta tangente buscada es pues

```
(%i8) y-0=3*(x-1);
```

$$(\%o8)\ y=3\,(x-1)$$

Si se quiere calcular la derivada segunda de la función definida implícitamente, hay que derivar implícitamente en la ecuación Eq2, y se obtiene

```
(%i9) diff(Eq2,x);
```

$$(\%o9)\ 3\,y^2\left(\frac{d^2}{d\,x^2}\,y\right)-x\left(\frac{d^2}{d\,x^2}\,y\right)+6\,y\left(\frac{d}{d\,x}\,y\right)^2-2\left(\frac{d}{d\,x}\,y\right)+6\,x=0$$

(%i10) Eq4:%;

$$(\%o10)\ 3\,y^2\left(\frac{d^2}{d\,x^2}\,y\right)-x\left(\frac{d^2}{d\,x^2}\,y\right)+6\,y\left(\frac{d}{d\,x}\,y\right)^2-2\left(\frac{d}{d\,x}\,y\right)+6\,x=0$$

Para calcular el valor de la derivada segunda en un punto, por ejemplo en *x*=1, hay que sustituir ahora los valores $x=1, y(1)=0, Dy(1)=3$ resulta:

(%i11) Eq5: -('diff(y,x,2))-2*3+6=0;

$$(\%o11)\ -\frac{d^2}{d\,x^2}\,y=0$$

Finalmente pues:

(%i12) 'D2y(1)=0;

(%o12) D2y(1)=0